


Fancier plotting for Math 240 -- fieldplot, spacecurve, and parametric surfaces

This section concerns two specialized plotting functions (in fact, they are commands from Maple's "plots" library) and an advanced use of the `plot3d` command for plotting surfaces.

1. `spacecurve` - is the Maple command used for plotting curves in three-dimensional space. In 3D, curves can be obtained by intersecting surfaces, but it is much more common to give parametrized versions of curves. For example, an example of a line in three dimensions is given by the parametric equations: $x=3+2t$, $y=4+t$, $z=1-t$. To draw the line with Maple, using `spacecurve`, you proceed as follows:

```
> with(plots,spacecurve);  
  
[spacecurve]  
> spacecurve([3+2*t,4+t,1-t],t=-3..3);
```



To get axes, etc. in a `spacecurve` plot, one clicks on the plot and then turns the plot box and uses the buttons on the plot command bar exactly as with `plot3d` (see the discussion there for more details). It is very common to want to combine a `spacecurve` plot with a plot of a surface. To combine 3D plots, one uses the

`display3d` command just as with combination of surface plots (again, see the section of this manual on `plot3d` for more detail). For example, here is a plot of a surface and one of the normal lines to it:

```
> F:=plot3d(x^2+y^2,x=-2..2,y=-2..2):
```

The normal line to this surface at the point (1,1,2) is given by parametric equations:

```
> normalline:=[1-2*t,1-2*t,2+t];
```


```
normalline := [1 - 2 t, 1 - 2 t, 2 + t]
```

```
> G:=spacecurve(normalline,t=-2..2):
```

```
> with(plots,display3d);
```

```
[display3d]
```

```
> display3d({F,G});
```



2. `fieldplot` - is used to draw pictures of vector fields in two dimensions. For example, let's consider the vector field $x\mathbf{i}+y\mathbf{j}$ --- in Maple we write this as:

```
> vfield:=[x,y];
```


```
vfield := [x, y]
```

To draw it, we enter:

```
> with(plots,fieldplot);
```

[fieldplot]

```
> fieldplot(vfield,x=-1..1,y=-1..1);
```




It is possible to combine plots of vector fields with plots of curves in the plane (to get a geometric idea of $F \cdot dr$ for example) using the `display` command. This works just as outlined for curves in the plot section of this manual, so we do not illustrate it here.

It is also possible to draw three-dimensional vector fields, with a command called `fieldplot3d`. Here is an example:

```
> with(plots,fieldplot3d); vectorfield:=[x,y,-2*z]:  
  fieldplot3d(vectorfield,x=-1..1,y=-1..1,z=-1..1);
```

[fieldplot3d]




Pictures of 3d vectorfields can be difficult to understand -- it is sometime instructive to watch as they are drawn by Maple. Again, it is possible to use the 3d plotting options to change perspectives, add axes, etc.. Also, one can combine a `fieldplot3d` with the plot of a curve or surface using `display3d` .

3. Parametric surfaces. It is often necessary when computing surface integrals to parametrize the surface using variables other than x , y or z . Maple (as part of the `plot3d` command) can draw parametrized surfaces, and this is useful sometimes for visualization and to make sure you've got the parametrization right. There are some examples of this in the "solved problems" section of this manual. Here is another -- note that the pieces of the parametrization go into obvious places, very analogous to parametric plotting of curves in 2 and 3 dimensions (discussed above and elsewhere in this manual). Here is the sphere of radius 2 centered at the point $(0,0,1)$, which is given parametrically (see the section of the text on spherical coordinates) by:

$x=2*\cos(u)*\cos(v)$, $y=2*\cos(u)*\sin(v)$, $z=1+2*\sin(u)$ as u goes from $-\pi/2$ to $\pi/2$ and v goes from 0 to $2*\pi$:

```
> plot3d([2*cos(u)*cos(v),2*cos(u)*sin(v),1+2*sin(u)],
          u=-Pi/2..Pi/2,v=0..2*Pi);
```



One thing to notice about the syntax is that (as opposed to parametric plotting of curves), the range expressions $u=-\pi/2.. \pi/2$ and $v=0..2*\pi$ go *outside* of the square brackets rather than inside.